```
(* RT *)
(* Lagrage's EOM *)
Apply [Clear,Names["Global`*"]];
Off[General::spell];
Off[General::spell1];
" polar coordinates - mobile reference frame "
rC= {L/2, 0, 0} ;
rA= {r[t], 0, 0} ;
Print["rC = ",rC];
Print["rA = ",rA];
G1= {-m g Sin[theta[t]], -m g Cos[theta[t]], 0};
G2= {-m g Sin[theta[t]], -m g Cos[theta[t]], 0};
Print["G1 = G2 = ", G1];
omega= {0, 0, theta'[t]};
vA=D[rA,t]+Cross[omega,rA];
vC=D[rC,t]+Cross[omega,rC];
Print["vA = ", vA];
Print["vC = ", vC];
IO=m L^2/3;
"T1= IO omega.omega/2 "
T1= IO omega.omega/2;
"T2= m vA.vA/2+IA omega.omega/2 "
T2=  m vA.vA/2+IA omega.omega/2;
T=T1+T2;
Print["T = T1 + T2 = ",Simplify[T]];

" left hand side of the Lagrange's equations "
"d[dT/d(r')]/dt - dT/dr = "
LHSr=D[D[T,r'[t]],t]-D[T,r[t]]
"d[dT/d(theta')]/dt - dT/d(theta) ="
LHSt=D[D[T,theta'[t]],t]-D[T,theta[t]]

"generalized forces"
"because is a mobile reference frame do not use"
"Qr = [d(rC)/dr].G1+[d(rA)/dr].G2  DO NOT USE"
"USE"
"Qr = [d(vC)/d(r')].G1+[d(vA)/d(r')].G2 = "
Qr=D[vC,r'[t]].G1+D[vA,r'[t]].G2
"Qt = [d(vC)/d(theta')].G1+[d(vA)/d(theta')].G2 ="
Qt=D[vC,theta'[t]].G1+D[vA,theta'[t]].G2

"Lagrange's EOM"
"d[dT/d(r')]/dt - dT/dr = Qr "
eq1=Simplify[LHSr-Qr]==0
"d[dT/d(theta')]/dt - dT/d(theta) = Qt"
eq2=Simplify[LHSt-Qt]==0

"numerical application"
rule={m→1.,L→1,IA→1.,g→10.};
equation1=Simplify[eq1/.rule]
equation2=Simplify[eq2/.rule]
sol=NDSolve[{equation1,equation2,r[0]==.1,theta[0]==.1,
        r'[0]==0.,theta'[0]==0.},{r,theta},{t,0.,1.}];
Plot[Evaluate[r[t]]/.sol,{t,0.,1.},PlotRange→All,
    AxesLabel→{"t[s]","r[m]"}];
Plot[Evaluate[theta[t]]/.sol,{t,0.,1.},PlotRange→All,
  AxesLabel→{"t[s]","theta[rad]"}]

 r5= (Evaluate[r[t]]/.sol/.t->.5)[[1]];
 t5= (Evaluate[theta[t]]/.sol/.t->.5)[[1]];
Print["r[.5] = ",r5, " m"];
Print["theta[.5] = ",t5, " rad"];
```

```
 polar coordinates - mobile reference frame
```

$rC = \{\frac{L}{2}, 0, 0\}$

$rA = \{r[t], 0, 0\}$

$G1 = G2 = \{-g\, m\, Sin[theta[t]], -g\, m\, Cos[theta[t]], 0\}$

$vA = \{r'[t], r[t]\, theta'[t], 0\}$

$vC = \{0, \frac{1}{2} L\, theta'[t], 0\}$

T1= IO omega.omega/2

T2= m vA.vA/2+IA omega.omega/2

$T = T1 + T2 = \frac{1}{6} (3\, m\, r'[t]^2 + (3\, IA + L^2\, m + 3\, m\, r[t]^2)\, theta'[t]^2)$

left hand side of the Lagrange's equations

d[dT/d(r')]/dt - dT/dr =

$-m\, r[t]\, theta'[t]^2 + m\, r''[t]$

d[dT/d(theta')]/dt - dT/d(theta) =

$2\, m\, r[t]\, r'[t]\, theta'[t] + IA\, theta''[t] + \frac{1}{3} L^2\, m\, theta''[t] + m\, r[t]^2\, theta''[t]$

generalized forces

because is a mobile reference frame do not use

Qr = [d(rC)/dr].G1+[d(rA)/dr].G2 DO NOT USE

USE

Qr = [d(vC)/d(r')].G1+[d(vA)/d(r')].G2 =

$-g\, m\, Sin[theta[t]]$

Qt = [d(vC)/d(theta')].G1+[d(vA)/d(theta')].G2 =

$-\frac{1}{2} g\, L\, m\, Cos[theta[t]] - g\, m\, Cos[theta[t]]\, r[t]$

Lagrange's EOM

d[dT/d(r')]/dt - dT/dr = Qr

$m\, (g\, Sin[theta[t]] - r[t]\, theta'[t]^2 + r''[t]) == 0$

d[dT/d(theta')]/dt - dT/d(theta) = Qt

$$\frac{1}{2} g\, L\, m\, Cos[theta[t]] + m\, r[t]\, (g\, Cos[theta[t]] + 2\, r'[t]\, theta'[t]) + \left(IA + \frac{L^2\, m}{3}\right) theta''[t] + m\, r[t]^2\, theta''[t] == 0$$

numerical application

```
10. Sin[theta[t]] + 1. r''[t] == 1. r[t] theta'[t]^2

5. Cos[theta[t]] + r[t] (10. Cos[theta[t]] + 2. r'[t] theta'[t]) +
  1.33333 theta''[t] + 1. r[t]^2 theta''[t] == 0
```






```
- Graphics -

r[.5] = 0.0971902 m

theta[.5] = -0.444956 rad
```